#qt引用模块核心功能
QT += core gui sql

#CONFIG += console

 
#这是Qt5为了兼容Qt4而专门设计的
greaterThan(QT_MAJOR_VERSION, 4): QT += widgets
 
#工程所使用的模版;app表示是应用程序;lib则表明是库
TEMPLATE=app
 
#临时文件存放位置
MOC_DIR         = temp/moc  #指定moc命令将含Q_OBJECT的头文件转换成标准.h文件的存放目录
RCC_DIR         = temp/rcc  #指定rcc命令将.qrc文件转换成qrc_*.h文件的存放目录
UI_DIR          = temp/ui   #指定rcc命令将.qrc文件转换成qrc_*.h文件的存放目录
OBJECTS_DIR     = temp/obj  #指定目标文件(obj)的存放目录
 
#指定生成的应用程序放置的目录
DESTDIR         = bin

SOURCES += \
    messagetips.cpp \
    widgets/ad_professionalsmgrlisr.cpp \
    csvtools.cpp \
    publicdataclass.cpp \
    widgets/ad_coursemgrlist.cpp \
    widgets/ST_onestudentinfo.cpp \
    widgets/ad_classmgrlist.cpp \
    widgets/ad_oneclassinfo.cpp \
    widgets/ad_studentmgrlist.cpp \
    widgets/ad_userinfo.cpp \
    widgets/ad_usermgrlist.cpp \
    dbtool.cpp \
    globaldata.cpp \
    widgets/log.cpp \
    main.cpp \
    mainwindow.cpp \
    pagejump.cpp \
    publictool.cpp \
    singletonroot.cpp \
    publicWidgets/topmenubtns.cpp \
    factorymodel.cpp \
    widgets/mytablevive.cpp \
    widgets/te_classgrade.cpp \
    widgets/topmenuadmin.cpp

HEADERS += \
    messagetips.h \
    widgets/ad_professionalsmgrlisr.h \
    csvtools.h \
    publicdataclass.h \
    widgets/ad_coursemgrlist.h \
    widgets/ST_onestudentinfo.h \
    widgets/ad_classmgrlist.h \
    widgets/ad_oneclassinfo.h \
    widgets/ad_studentmgrlist.h \
    widgets/ad_userinfo.h \
    widgets/ad_usermgrlist.h \
    dbtool.h \
    globaldata.h \
    widgets/log.h \
    mainwindow.h \
    pagejump.h \
    publictool.h \
    singletonroot.h \
    publicWidgets/topmenubtns.h \
    factorymodel.h \
    widgets/mytablevive.h \
    widgets/te_classgrade.h \
    widgets/topmenuadmin.h

FORMS += \
    widgets/ad_professionalsmgrlisr.ui \
    widgets/ad_coursemgrlist.ui \
    widgets/ST_onestudentinfo.ui \
    widgets/ad_classmgrlist.ui \
    widgets/ad_oneclassinfo.ui \
    widgets/ad_studentmgrlist.ui \
    widgets/ad_userinfo.ui \
    widgets/ad_usermgrlist.ui \
    widgets/log.ui \
    mainwindow.ui \
    publicWidgets/topmenubtns.ui \
    widgets/mytablevive.ui \
    widgets/te_classgrade.ui \
    widgets/topmenuadmin.ui
 
#指定生成的应用程序放置的目录
IDE_SOURCE_TREE = $$PWD #.pro或.pri文件所在的位置
IDE_BUILD_TREE  = $$IDE_SOURCE_TREE/../
win32:CONFIG(debug, debug|release){
    contains(DEFINES, WIN64) {
    DESTDIR = $$IDE_BUILD_TREE/_debug64
    } else {
    DESTDIR = $$IDE_BUILD_TREE/_debug86
    }
} else:win32:CONFIG(release, debug|release){
    contains(DEFINES, WIN64) {
    DESTDIR = $$IDE_BUILD_TREE/_release64
    } else {
    DESTDIR = $$IDE_BUILD_TREE/_release86
    }
}

macx:CONFIG(debug, debug|release){
    DESTDIR = $$IDE_BUILD_TREE/_debug64
} else:macx:CONFIG(release, debug|release){
    DESTDIR = $$IDE_BUILD_TREE/_release64
}
 
#指定生成的应用程序名和图标
# TARGET   = Hello
# RC_ICONS = Hello.ico

#{TARGET}
#{RC_ICONS}
 
#定义编译选项
#QT_DEPRECATED_WARNINGS表示当Qt的某些功能被标记为过时的,那么编译器会发出警告
DEFINES += QT_DEPRECATED_WARNINGS
 
#指定编译器选项和项目配置
CONFIG  += c++11
CONFIG  += warn_on           #告诉qmake要把编译器设置为输出警告信息的
# CONFIG  += precompile_header #可以在项目中使用预编译头文件的支持

INCLUDEPATH += $$PWD/widgets $$PWD/publicWidgets

#预编译头文件路径
PRECOMPILED_HEADER = $$PWD/singletonroot.h
 

msvc {
    QMAKE_CFLAGS += /utf-8
    QMAKE_CXXFLAGS += /utf-8
}


message("end")

win32:CONFIG(release, debug|release): LIBS += -L$$PWD/qtcsv/lib/ -lqtcsv
else:win32:CONFIG(debug, debug|release): LIBS += -L$$PWD/qtcsv/lib/ -lqtcsvd
else: LIBS += -L$$PWD/qtcsv/lib/ -lqtcsv1


INCLUDEPATH += $$PWD/qtcsv/include
DEPENDPATH += $$PWD/qtcsv/include
